

Local Perturbation Analysis (LPA) Tutorial using XPP

WR Holmes, May Anne Mata,
And Leah Edelstein-Keshet

Aug, 2014

Local Perturbation Analysis (LPA)

- LPA is a reduction technique that approximates the initial behaviour of a very narrow but arbitrarily tall pulse used as input to a reaction-diffusion system. The method was devised by AFM Marée.
- When the rates of diffusion of intermediates differ widely, the system can be approximated by local (pulse height) and global (background levels) of the variables.
- This approximation reduces RD PDEs to a set of ODEs for the local and global variables; these can then be studied with available ODE bifurcation methods.
- Here we illustrate the use of XPPaut to construct such bifurcation plots.

Installation and basic XPPaut

XPPaut is a freely available ODE solver linked to the Auto bifurcation software. It was created by G.B. Ermentrout, who maintains and updates the software. To download the program and install it, and to see an online tutorial visit

<http://www.math.pitt.edu/~bard/xpp/xpp.html>

A guide to XPP with many examples is available in:

G B Ermentrout (2002) *Simulating, Analyzing, and Animating Dynamical Systems: A Guide to XPPAUT*, SIAM, Philadelphia

See also an introductory modeling text with many XPP examples:

Lee A Segel and L Edelstein-Keshet (2013) *A Primer on Mathematical Biology*, SIAM, Philadelphia.

Here we assume a very basic familiarity with XPP in explaining the LPA bifurcation technique.

Example 1: Schnakenberg RD and LPA

- The Schnakenberg system is a well known pattern-forming reaction diffusion system. In PDE form it is

$$u_t(x, t) = a - u + u^2v + D_u\Delta u$$

$$v_t(x, t) = b - u^2v + D_v\Delta v,$$

- Here $D_u \ll D_v$ so u will be represented by both a local and a global variable. The LPA equations are

$$u_L' = a - u_L + u_L^2 v_G$$

$$u_G' = a - u_G + u_G^2 v_G$$

$$v_G' = b - u_G^2 v_G$$

Schnakenberg LPA XPP .ode file

```
# SchnakenLPA.ode
```

```
#Define the expressions in the reaction-diffusion system
```

```
f(u,v)=a-u+(u^2)*v
```

```
g(u,v)=b-(u^2)*v
```

```
# Write the LPA ODES for local and global variables
```

```
ul'=f(ul,vg)
```

```
ug'=f(ug,vg)
```

```
vg'=g(ug,vg)
```

```
# Optionally set initial conditions for the variables
```

```
init ul=0,ug=3,vg=0.1
```

```
# Define a basic set of parameter values, specify the axes to plot and time to integrate
```


```
par a=0.5,b=1
```

```
@ xp=t,yp=ul,xlo=0,ylo=0,xhi=100,yhi=6,total=100
```

```
done
```


The XPP window

- Starting XPP brings up a window and menu




Displays

- Optionally clicking on the small buttons “Param” and “ICs” will bring up windows with the default sets of parameters and initial conditions





Integrate the system

- From the main window, select
- Initial conditions, **Go** (or type **I G**) to integrate the equations.
- The solution for u_L appears in the window




Get close to a steady state (SS)

- From the main window, select **Initial conditions, Last** (or type **I L**); repeat several times to get very close to steady state.
- Initial Data will approach SS
- Solution for UL will be flat




Start Auto

- From the main window select **File**, **Auto** or type **F A** to bring up an AUTO bifurcation window:



Settings and panels


- Select bifurcation parameter
- Scale axes
- Technical settings
- Run continuation
- Grab a point
- Eigenvalues



Parameter


- Auto selects the first parameter in the .ode file as its bifurcation parameter. Either change to your desired parameter or click OK

- Parameter a is the bifurcation parameter in this example.




Scale the axes

- Click Axes
- Select variable to plot (u_L)
- Chose scaling
- Click OK
- Plot will be relabeled and Rescaled accordingly
- Scaling can later be changed



Technical settings (Numerics)


- **Step size** (adjust later to + or – direction)
- **Min value** (adjust to 0.5)
- **Max value**



Leave other settings as is for most purposes. In some cases, adjustments needed to the tolerance and step size.


Run

- Click **Run**, select **Steady State**
- The following diagram will appear in the AUTO window:
- Numbered points can be selected by **Grab**.



Complete the diagram

- Click **Grab**; a point is highlighted by a large **+** ; use the tab button to skip between numbered points on the bifurcation plot.
- To select desired point (1) click **Return**
- Adjust direction of continuation





Continue in decreasing direction

Ntst:15	Dsmax:0.5
Nmax:200	Par Min:0
NPr:50	Par Max:2
Ds:-0.02	Norm Min:0

Continue all the way to a=0

Complete the diagram


- Diagram obtained so far shown above. Now we complete the branch ending in point labeled 5.
- Click **Grab**
- Click **Tab** to get to point 5
- **Return**
- **Run**
- The diagram is ready:



Two regimes of behavior:


- LPA diagram reveals two regimes of distinct behavior:

- Spontaneous patterns
- Stimulus-induced patterns






Print to postscript file

- Select **F**ile; **P**ostscript
- Specify name of file to save
- A diagram will be saved



Eigenvalues

- Tabbing through diagram reveals real and imaginary parts of eigenvalues



Example 2: Gierer-Meinhardt RD


A pattern-forming system with periodic “flashing” patterns is the following GM system:

$$u_t(x, t) = a - bu + \frac{u^2}{v(1 + Ku^2)} + \Delta u$$

$$v_t(x, t) = u^2 - v + D\Delta v.$$

Here $D \gg 1$, so u will be represented by a local LPA variable.

See LPA equations and .ode file next page.



XPP .ode file for the Gierer-Meinhard system

```
# GeirerMeinhLPA.ode
```

```
f(u,v)=a-b*u+u^2/(v*(1+k*u^2))
```

```
g(u,v)=u^2-v
```

```
ul'=f(ul,vg)
```

```
ug'=f(ug,vg)
```

```
vg'=g(ug,vg)
```

```
init ul=0.2,ug=1,vg=1
```



```
par a=0.5,b=2,k=0
```

```
@ xp=t,yp=ul,xlo=0,ylo=0,xhi=100,yhi=3,total=100
```

```
done
```


Similarly run the system and create auto bifurcation diagram

- Run with default settings: produces this diagram
- Reversing direction of continuation ($ds=-0.02$) completes the branch
- **Grab point 9 (HB); Run, Periodic**



Hopf Bifurcation


- Hopf bifurcation shown at point 9 ($a=0.333$), indicates the presence of a limit cycle in the LPA system



Three regimes of behavior

- LPA system reveals three regimes of distinct behavior:


- (I) Time-Periodic Spatial patterns
- (II) Spontaneous static patterns
- (III) Stimulus-induced pattern formation



Example 3: Rac-Rho mutual inhibition

GTPases Rac and Rho mutually inhibit. Active forms diffuse slowly in the cell membrane, inactive forms diffuse fast in the cytosol. Total (active + inactive) of each GTPase is constant.

Model contains 4 equations for active and inactive forms of each GTPase.



Rac-Rho mutual inhibition system

- Reaction-diffusion equations for active forms:

$$\frac{\partial R}{\partial t} = \left(k_R + \frac{I_R}{1 + \left(\frac{\rho}{a_1} \right)^n} \right) \frac{R_i}{R_t} - \delta R + D_R \Delta R$$
$$\frac{\partial \rho}{\partial t} = \left(k_\rho + \frac{I_\rho}{1 + \left(\frac{R}{a_2} \right)^n} \right) \frac{\rho_i}{\rho_t} - \delta \rho + D_\rho \Delta \rho$$

- Two more PDEs for inactive forms R_i, ρ_i
- Same kinetic terms (but of opposite signs) and faster (cytosolic) rates of diffusion.

Conservation of total GTPase

- Total amounts of active and inactive forms, $R+R_i$, $\rho+\rho_i$ conserved in the domain.
- Mass-conservative system is degenerate (has a zero eigenvalue), leading to problem for continuation software.
- Cure: eliminate the inactive forms in the LPA ODES, using conservation:

$$R_i = R_{total} - R, \quad \rho_i = \rho_{total} - \rho,$$

Rac-Rho LPA system and .ode file

```
# RacRhoLPA.ode
```

```
# LPA for mutual inhibitory Rac|--|Rho system
```

```
# Kinetic terms
```

```
f(R,rho,Rg)=(kr+lr/(1+(rho/a1)^n))*(Ri(Rg))-dr*R
```

```
g(R,rho,rhog)=(krho+lrho/(1+(R/a2)^n))*(rhoi(rhog))-drho*rho
```

```
# Inactive forms of GTPases (as fraction of total amount)
```

```
Ri(Rg)= 1-Rg/R_total
```

```
rhoi(rhog)=1-rhog/rho_total
```

```
# LPA ODEs for the local and global variables
```

```
Rl'=f(Rl,rhol,Rg)
```

```
Rg'=f(Rg,rhog,Rg)
```

```
rhol'=g(Rl,rhol,rhog)
```

```
rhog'=g(Rg,rhog,rhog)
```

```
# Optional initial conditions
```

```
init Rg=0.8, rhog=0.8, Rl=1,
```

```
par kr=0.05, krho=0.2, drho=1, dr=1, lr=1, lrho=1
```

```
par R_total=1, rho_total=1, a1=0.25, a2=0.25, n=3
```

```
@ xp=t,yp=Rl,xlo=0,ylo=0,xhi=100,yhi=1,total=100
```

```
done
```

Kinetic terms in PDEs


Conservation used to calculate inactive fraction

LPA ODEs for local and global variables

Parameter values and various settings (kr will be the default bifurcation parameter)


Integrating the system

- Running XPP and typing **I G, I L, I L** produces approach to steady state in the local variable R_L



Rac-Rho LPA: AUTO bifurcation

- Type **F A**. To bring up AUTO
- Set **Axes** to hI-Lo, set Xmin:0, Ymin:0, Xmax:1, Ymax:1
- **Numerics:**




The screenshot shows a dialog box titled "AutoNum" with the following settings:

Ntst:15	Dsmax:0.5
Nmax:200	Par Min:0
NPr:50	Par Max:1
Ds:0.001	Norm Min:0



- **Run, Steady state**
- Auto will complete entire diagram on its own in this case.


Bifurcation diagram

- You should obtain the following diagram:



Four regimes of behavior:

- LPA diagram predicts four regimes:
- Upstimulus induces pattern 
- Spontaneous pattern forms (for arbitrarily small noise)
- Down stimulus induces pattern 
- No pattern forms



Common troubleshooting

- Be sure to start at a steady state of the system by integrating long enough to see no further change.
- If auto fails to find several branches, consider starting at a different parameter value and initial conditions to find different steady state.

Further references

- Jilkine A, Marée AFM, Edelstein-Keshet L (2007) Mathematical model for spatial segregation of the Rho-family GTPases based on inhibitory crosstalk, *Bulletin for Mathematical Biology*, 69(6): 1943-78.
- Holmes WR, Lin B, Levchenko A, Edelstein-Keshet L (2012) Modeling cell polarization driven by synthetic spatially graded Rac activation, *PLoS Computational Biology* 8(6): e1002366.
- Walther G R, Marée AFM, Edelstein-Keshet L, Grieneisen V (2012) Deterministic versus stochastic cell polarization through wave pinning, *Bull Math Biol* 74 (11), 2570-2599.
- Holmes WR, Carlsson A, Edelstein-Keshet L (2012) Regimes of wave type patterning driven by refractory actin feedback: transition from static polarization to dynamic wave behaviour, *Phys Biol.* 9, 046005
- Holmes WR (2014) An efficient, nonlinear stability analysis for detecting pattern formation in reaction diffusion systems, *Bull math biol* 76: 157-183.
- Mata MA, Dutot M, Edelstein-Keshet L, Holmes WR (2013) A model for intracellular actin waves explored by nonlinear local perturbation analysis, *J theor Biol.* 334: 149–161.